Journal of Software for Algebra and Geometry

Free resolutions and modules with a semisimple Lie group action

Federico Galetto

vol 7 2015

JSAG **7** (2015), 17–29
dx.doi.org/10.2140/jsag.2015.7.17

The Journal of Software for
Algebra and Geometry

Free resolutions and modules with a semisimple Lie group action

FEDERICO GALETTO

ABSTRACT: We introduce the package *HighestWeights* for Macaulay2. This package provides tools to study the representation-theoretic structure of free resolutions and graded modules over a polynomial ring with the action of a semisimple Lie group. The methods of this package allow users to consider the free modules in a resolution, or the graded components of a module, as representations of a semisimple Lie group by means of their weights, and to obtain their decomposition into highest-weight representations.

1. INTRODUCTION. Let R be a polynomial ring over the complex numbers with a $\mathbb{Z}^m$-grading, and let M be a finitely generated graded R-module. Under mild assumptions on R, for every degree $d \in \mathbb{Z}^m$ each graded component M_d is a finite-dimensional complex vector space. Next assume that G is a complex Lie group and that there is a degree-preserving $\mathbb{C}$-linear action of G on R compatible with multiplication, i.e., such that for every $g \in G$, $r_1, r_2 \in R$ we have $g \cdot (r_1 r_2) = (g \cdot r_1)(g \cdot r_2)$. We are interested in those R-modules M with a degree-preserving $\mathbb{C}$-linear action of G compatible with the module structure, i.e., such that for every $g \in G$, $r \in R$ and $m \in M$ we have $g \cdot (rm) = (g \cdot r)(g \cdot m)$. Notice that each graded component of such a module M is stable under the action of G.

Examples of such modules can arise naturally. For instance, let X be a finite-dimensional representation of a complex Lie group G. The symmetric algebra $\mathrm{Sym}(X)$, with the standard grading determined by $\mathrm{Sym}^1(X) \cong X$, is an example of a polynomial ring with a degree-preserving $\mathbb{C}$-linear action of G. The action of G also extends to the projective space $\mathbb{P}(X)$. If V is a projective variety in $\mathbb{P}(X)$ which is fixed by the action of G, then the affine cone $\hat{V}$ is an affine variety in X which is fixed by G. Moreover, $\hat{V}$ is the zero locus of some radical homogeneous ideal I in $\mathrm{Sym}(X)$, and the quotient ring $\mathrm{Sym}(X)/I$, i.e., the affine coordinate ring of $\hat{V}$, is an example of a $\mathrm{Sym}(X)$-module with a compatible G-action.

MSC2010: primary 13P20; secondary 22E46.
Keywords: equivariant free resolution, irreducible representation, highest weight, algebraic torus, semisimple Lie group, decomposition algorithm.

Let M be a finitely generated graded R-module with the kind of G-action described earlier. Denote by $\mathfrak{m}$ the maximal ideal generated by the variables in R, and set $V_0 = M/\mathfrak{m}M$. The vector space V_0 is a finite-dimensional graded representation of G. Assuming G is linearly reductive, the natural projection $M \to V_0$ admits a section $V_0 \to M$ which is compatible with the grading and G-action. This section extends to an R-linear map $d_0 : F_0 \to M$, where F_0 is defined to be $V_0 \otimes_{\mathbb{C}} R$. By construction, F_0 is a graded free R-module with a natural action of G that commutes with d_0; moreover, d_0 maps a basis of F_0 to a minimal generating set of M. Notice also that $F_0/\mathfrak{m}F_0 \cong V_0$ as a graded representation of G. Next, let $V_1 = \ker(d_0)/\mathfrak{m}\ker(d_0)$. Using the same ideas as before, let F_1 be $V_1 \otimes_{\mathbb{C}} R$ (so that $F_1/\mathfrak{m}F_1 \cong V_1$) and define the map $d_1 : F_1 \to F_0$. Again, d_1 will be a map of graded free R-modules that commutes with the action of G and maps a basis of F_1 to a minimal generating set of $\ker(d_0)$; in addition, $d_1 \circ d_0 = 0$. Iterating this procedure constructs a minimal free resolution of M with a built-in action of G; in particular, the procedure is guaranteed to terminate. Since any two minimal free resolutions of M are isomorphic as complexes, the action of G transfers to every other minimal free resolution via the isomorphism with the one just constructed.

Summarizing what we said so far, for every finitely generated graded R-module M with a degree-preserving G-action compatible with the module structure, the following occurs:

(1) For every degree $d \in \mathbb{Z}^m$, the graded component M_d is a finite-dimensional representation of G.

(2) If the complex

$$0 \longleftarrow M \longleftarrow F_0 \longleftarrow F_1 \longleftarrow \cdots \longleftarrow F_n \longleftarrow 0$$

is a minimal free resolution of M as an R-module, then the action of G on M extends to each F_i and there is an isomorphism $F_i \cong (F_i/\mathfrak{m}F_i) \otimes_{\mathbb{C}} R$ of graded R-modules with a G-action. Each $F_i/\mathfrak{m}F_i$ is a finite-dimensional graded vector space so, for each degree $d \in \mathbb{Z}^m$, $(F_i/\mathfrak{m}F_i)_d$ is a finite-dimensional representation of G.

When the group G is semisimple, it is a typical problem to decompose a finite-dimensional representation into irreducible representations. Moreover each irreducible representation is indexed by a so-called highest weight. The main purpose of *HighestWeights* is to provide users of Macaulay2 [Grayson and Stillman] with tools to obtain the highest-weight decomposition of the representations M_d and $(F_i/\mathfrak{m}F_i)_d$ introduced above. This purpose is achieved by implementing an algorithm for propagating weights of tori along equivariant maps introduced in [Galetto 2015].

This article is organized as follows: the next section details the mathematical assumptions for using this package, Section 3 presents two examples in detail, and Section 4 contains some final remarks. We have also included a brief review of the representation theory of semisimple Lie groups in the Appendix.

2. PACKAGE ASSUMPTIONS. Before presenting some examples, we discuss some assumptions of this package. Using the notation of the introduction, the polynomial ring R must be positive $\mathbb{Z}^m$-graded for some positive integer m, in the sense of [Kreuzer and Robbiano 2005, Definition 4.2.4]. More explicitly, if $R = \mathbb{C}[x_1, \ldots, x_n]$, then R is graded by elements of $\mathbb{Z}^m$ in such a way that:

(1) each nonzero constant in R has degree $0 \in \mathbb{Z}^m$;

(2) the degree of each variable $x_i \in R$ is a nonzero vector in $\mathbb{Z}^m$, and its first nonzero entry is positive;

(3) the matrix with rows given by the degrees of the variables $x_1, \ldots, x_n$ has rank m.

This ensures that, for every finitely generated graded R-module M and for every degree $d \in \mathbb{Z}^m$, each graded component M_d is a finite-dimensional complex vector space. The variables in the polynomial ring R must be weight vectors for the action of the chosen maximal torus in G; this can always be achieved up to a linear change of variables in R. The user is expected to provide the weight of each variable. Any monomial ordering on the monomials of R is allowed. All free R-modules must be endowed with a term over position up/down or position up/down over term ordering; in Macaulay2 this is established with the declaration of the ring. The default ordering, term over position up, is fine for most computations, unless the user needs a different one.

To obtain the decomposition of a graded component M_d, the user is expected to provide a presentation $\varphi : F_1 \to F_0$ of M in the form of a matrix written with respect to a homogeneous basis $\{e_1, \ldots, e_r\}$ of F_0 such that the residue classes $\bar{e}_1, \ldots, \bar{e}_r$ modulo $\mathfrak{m}F_0$ form a basis of weight vectors of $F_0/\mathfrak{m}F_0$; the user will also need to provide a list with the weights of $\bar{e}_1, \ldots, \bar{e}_r$. For modules with a compatible group action, presentations of this kind are, in our experience, the most natural. As for resolutions, the user must provide a list of weights for a basis $e_1, \ldots, e_r$ as before, for any one of the modules F_i. When $M = R/I$, for a G-stable ideal I in R, the module F_0 is simply R with a trivial G-action; in this case, the user does not need to input any weight (other than those of the variables of R).

The package *WeylGroups*, which is loaded automatically by this package, is used to declare the type of a semisimple group and to handle many weight-related operations behind the scenes. However, for the purpose of this package, weights are to be provided simply as lists of integers, not as objects of type `Weight` as in *WeylGroups*.

3. Examples. For an explanation of the notations and conventions relating to weights that appear in the following examples, we refer the reader to Section A2.

3.1. ***The coordinate ring of the Grassmannian.*** Let $E = \mathbb{C}^6$, the standard representation of $\mathrm{SL}_6(\mathbb{C})$, with coordinate basis $\{e_0, \ldots, e_5\}$. The Grassmannian $V = \mathrm{Gr}(2, E^*)$ is the projective variety which parametrizes two-dimensional subspaces of E^*; it is embedded in $\mathbb{P}(\bigwedge^2 E^*)$ using the Plücker equations [Shafarevich 1994, Chapter I, §4.1]. Consider $\bigwedge^2 E^*$ as a complex affine space. Let C be the affine cone over V, i.e., the subvariety of $\bigwedge^2 E^*$ which is the union of all the one-dimensional subspaces of $\bigwedge^2 E^*$ belonging to V. The space $\bigwedge^2 E^*$ has a natural action of $\mathrm{SL}_6(\mathbb{C})$ which fixes C.

Our polynomial ring R is the ring of polynomial functions over $\bigwedge^2 E^*$, i.e., the symmetric algebra $\mathrm{Sym}(\bigwedge^2 E)$. The elements $p_{i,j} = e_i \wedge e_j$ for $0 \leqslant i < j \leqslant 5$ form a basis of weight vectors of $\bigwedge^2 E$ and will be the variables in R. The defining ideal of C is generated by the Plücker equations; this ideal, which we call I, can be conveniently obtained in Macaulay2 using the command `Grassmannian`. We resolve the quotient R modulo I as an R-module and call `RI` the minimal free resolution:

```
i1 : printWidth=72; truncateOutput 200;

i3 : I=Grassmannian(1,5,CoefficientRing=>QQ); R=ring I;

o3 : Ideal of QQ[p   , p   , p   , p   , p   , p   , p   , p   , p   , ... ]
                  0,1   0,2   1,2   0,3   1,3   2,3   0,4   1,4   2,4

i5 : RI=res I; betti RI

            0  1  2  3  4  5 6
o6 = total: 1 15 35 42 35 15 1
         0: 1  .  .  .  .  . .
         1: . 15 35 21  .  . .
         2: .  .  . 21 35 15 .
         3: .  .  .  .  .  . 1

o6 : BettiTally
```

Now we load the package and assign weights to the variables of R. First we input the weights of $e_0, \ldots, e_5$ in a list `L`.

```
i7 : loadPackage "HighestWeights";

i8 : L={{1,0,0,0,0},{-1,1,0,0,0},{0,-1,1,0,0},{0,0,-1,1,0},{0,0,0,-1,1},
        {0,0,0,0,-1}};
```

The weight of $p_{i,j} = e_i \wedge e_j$ is equal to the sum of the weights of e_i and e_j (see Appendix A2). The subscripts of the variables $p_{i,j}$ are the elements of `subsets({0,1,2,3,4,5},2)`, the 2-subsets of the set $\{0, 1, 2, 3, 4, 5\}$. Hence taking sums of pairs of weights in L over this indexing set will give us a complete list of weights for the variables $p_{i,j}$, as listed by Macaulay2:

```
i9 : W=apply(subsets({0,1,2,3,4,5},2),s->L_(s_0)+L_(s_1))

o9 = {{0, 1, 0, 0, 0}, {1, -1, 1, 0, 0}, {-1, 0, 1, 0, 0}, {1, 0, -1, 1, 0},
     ------------------------------------------------------------------------
     {-1, 1, -1, 1, 0}, {0, -1, 0, 1, 0}, {1, 0, 0, -1, 1}, {-1, 1, 0, -1, 1},
     ------------------------------------------------------------------------
     {0, -1, 1, -1, 1}, {0, 0, -1, 0, 1}, {1, 0, 0, 0,  ...

o9 : List
```

We declare D to be the Dynkin type A_5, which is the type of the group $\mathrm{SL}_6(\mathbb{C})$. We then attach the weights in W to the variables in R with the command `setWeights`; the arguments are the ring, the type and the weights of the variables, respectively. The output will be the highest-weight decomposition of the $\mathbb{C}$-linear subspace of R generated by its variables; it is given in the form of a `Tally`, with keys describing the highest weights of the irreducible representation appearing in the decomposition and values equal to the multiplicities of those representations. In this case, we get simply `{0, 1, 0, 0, 0} => 1`, which means that the decomposition contains only one copy of the irreducible representation with highest weight `{0, 1, 0, 0, 0}`, i.e., $\bigwedge^2 E$, as expected:

```
i10 : D=dynkinType{{"A",5}}; setWeights(R,D,W)

o11 = Tally{{0, 1, 0, 0, 0} => 1}

o11 : Tally
```

All monomials in R are weight vectors. To recover the weight of a monomial, use the command `getWeights` with the monomial as the argument:

```
i12 : getWeights(p_(0,1)*p_(1,2))

o12 = {-1, 1, 1, 0, 0}

o12 : List
```

We can now issue the command `highestWeightsDecomposition` to obtain the decomposition of the representations corresponding to the free modules in the resolution; the only argument is the resolution RI. Suppose the free modules in RI are $F_0, \dots, F_6$. The outermost `HashTable` in the output has keys equal to the subscripts of the free modules in RI. The value corresponding to a key i is itself a `HashTable` with keys equal to the degrees of the generators of F_i. Finally the value corresponding to a certain degree d is a `Tally` containing the highest-weight decomposition of the representation $(F_i/\mathfrak{m}F_i)_d$, as described earlier:

```
i13 : highestWeightsDecomposition(RI)

o13 = HashTable{0 => HashTable{{0} => Tally{{0, 0, 0, 0, 0} => 1}}}
                1 => HashTable{{2} => Tally{{0, 0, 0, 1, 0} => 1}}
                2 => HashTable{{3} => Tally{{1, 0, 0, 0, 1} => 1}}
```

```
                3 => HashTable{{4} => Tally{{2, 0, 0, 0, 0} => 1}}
                               {5} => Tally{{0, 0, 0, 0, 2} => 1}
                4 => HashTable{{6} => Tally{{1, 0, 0, 0, 1} => 1}}
                5 => HashTable{{7} => Tally{{0, 1, 0, 0, 0} => 1}}
                6 => HashTable{{9} => Tally{{0, 0, 0, 0, 0} => 1}}

o13 : HashTable
```

By analyzing this output, we obtain the following description for `RI`:

$$R \leftarrow \bigwedge^4 E \otimes R(-2) \leftarrow \mathbb{S}_{2,1,1,1,1} E \otimes R(-3) \\ \leftarrow \mathbb{S}_2 E \otimes R(-4) \oplus \mathbb{S}_{2,2,2,2,2} E \otimes R(-5) \leftarrow \mathbb{S}_{2,1,1,1,1} E \otimes R(-6) \\ \leftarrow \bigwedge^2 E \otimes R(-7) \leftarrow R(-9) \leftarrow 0.$$

Next we turn to the coordinate ring of C, i.e., the quotient ring $Q = R/I$. We decompose its graded components in the range of degrees from 0 to 4, again with the command `highestWeightsDecomposition`. This time the arguments are the ring followed by the lowest and highest degrees in the range to be decomposed:

```
i14 : Q=R/I; highestWeightsDecomposition(Q,0,4)

o15 = HashTable{0 => Tally{{0, 0, 0, 0, 0} => 1}}
                1 => Tally{{0, 1, 0, 0, 0} => 1}
                2 => Tally{{0, 2, 0, 0, 0} => 1}
                3 => Tally{{0, 3, 0, 0, 0} => 1}
                4 => Tally{{0, 4, 0, 0, 0} => 1}

o15 : HashTable
```

We deduce that $(R/I)_d = \mathbb{S}_{d,d} E$ for $d \in \{0, \ldots, 4\}$. We can also decompose the graded components of the ring R in a range of degrees or in a single degree:

```
i16 : highestWeightsDecomposition(R,2)

o16 = Tally{{0, 0, 0, 1, 0} => 1}
            {0, 2, 0, 0, 0} => 1

o16 : Tally
```

For example, $R_2 = \bigwedge^4 E \oplus \mathbb{S}_{2,2} E$. Since the representation $\bigwedge^4 E$ appears in R_2 but not in $(R/I)_2$, we deduce that it must be in I_2, the graded component of I of degree 2. This can be verified directly by decomposing I_2 as follows:

```
i17 : highestWeightsDecomposition(I,2)

o17 = Tally{{0, 0, 0, 1, 0} => 1}

o17 : Tally
```

3.2. ***The Buchsbaum–Rim complex.*** Let $E = \mathbb{C}^6$ with coordinate basis $\{e_1, \ldots, e_6\}$ and $F = \mathbb{C}^3$ with coordinate basis $\{f_1, f_2, f_3\}$. Denote by R the symmetric algebra $\mathrm{Sym}(E \otimes F)$; R is a polynomial ring with variables $x_{i,j} = e_i \otimes f_j$. We take M to be the cokernel of a generic 3×6 matrix of variables in R. The minimal free resolution of M is an example of a Buchsbaum–Rim complex [Eisenbud 1995, Appendix A2.6]. We call this complex `BR`.

```
i1 : printWidth = 72; truncateOutput 200;

i3 : R=QQ[x_(1,1)..x_(6,3)];

i4 : G=genericMatrix(R,3,6)

o4 = | x_(1,1) x_(2,1) x_(3,1) x_(4,1) x_(5,1) x_(6,1) |
     | x_(1,2) x_(2,2) x_(3,2) x_(4,2) x_(5,2) x_(6,2) |
     | x_(1,3) x_(2,3) x_(3,3) x_(4,3) x_(5,3) x_(6,3) |

              3       6
o4 : Matrix R  <--- R

i5 : M=coker G; BR=res M; betti BR

            0 1  2  3 4
o7 = total: 3 6 15 18 6
         0: 3 6  .  . .
         1: . .  .  . .
         2: . . 15 18 6

o7 : BettiTally
```

The ring R carries a degree compatible action of $SL_6(\mathbb{C}) \times SL_3(\mathbb{C})$. Define the map of graded free R-modules

$$\varphi : E \otimes R(-1) \to F^* \otimes R, \quad e_i \otimes 1 \mapsto \sum_{j=1}^{3} f_j^* \otimes x_{i,j},$$

where $\{f_1^*, f_2^*, f_3^*\}$ is the dual basis in F^*. The matrix of φ with respect to the bases $\{e_1 \otimes 1, \ldots, e_6 \otimes 1\}$ and $\{f_1^* \otimes 1, f_2^* \otimes 1, f_3^* \otimes 1\}$ is precisely the generic matrix G introduced above. Moreover, φ is $SL_6(\mathbb{C}) \times SL_3(\mathbb{C})$-equivariant, meaning that for all $g \in SL_6(\mathbb{C}) \times SL_3(\mathbb{C})$, $e \in E$ and $r \in R$ we have $\varphi(g \cdot (e \otimes r)) = g \cdot \varphi(e \otimes r)$. This makes its cokernel M a module with a compatible $SL_6(\mathbb{C}) \times SL_3(\mathbb{C})$-action.

The weight of $x_{i,j} = e_i \otimes f_j$ is obtained by concatenating the weight of e_i with that of f_j. First we record the weights of $e_1, \ldots, e_6$ in a list e and those of f_1, f_2, f_3 in a list `f`. Then we concatenate them as illustrated below and attach the resulting list to the variables $x_{i,j}$. Care must be taken that the order of the weights matches the order of the variables:

```
i8 : loadPackage "HighestWeights";

i9 : e={{1,0,0,0,0},{-1,1,0,0,0},{0,-1,1,0,0}, {0,0,-1,1,0},{0,0,0,-1,1},
        {0,0,0,0,-1}};

i10 : f={{1,0},{-1,1},{0,-1}};

i11 : W=flatten table(e,f,(u,v)->u|v)

o11 = {{1, 0, 0, 0, 0, 1, 0}, {1, 0, 0, 0, 0, -1, 1}, {1, 0, 0, 0, 0, 0, -1},
      ------------------------------------------------------------------------
      {-1, 1, 0, 0, 0, 1, 0}, {-1, 1, 0, 0, 0, -1, 1}, {-1, 1, 0, 0, 0, 0, -1},
      ------------------------------------------------------------------------
      {0, -1, 1, 0, 0, 1, 0}, {0, -1, 1, 0, 0, -1, 1}, { ...

o11 : List

i12 : D=dynkinType{{"A",5},{"A",2}}; setWeights(R,D,W)
```

```
o13 = Tally{{1, 0, 0, 0, 0, 1, 0} => 1}
o13 : Tally
```

In order to decompose the representations in a resolution, we need to ensure that the coordinate basis for at least one of the free modules in the resolution is a basis of weight vectors, and then we need to input the weights of the elements of that basis. For our resolution BR, we could choose the first or the second free module. In fact, the first differential of BR is the map $\varphi : E \otimes R(-1) \to F^* \otimes R$ whose matrix was written with respect to the bases of weight vectors $\{e_1 \otimes 1, \dots, e_6 \otimes 1\}$ and $\{f_1^* \otimes 1, f_2^* \otimes 1, f_3^* \otimes 1\}$. We choose to work with the first module, i.e., the codomain of φ. Notice that the element $1 \in R$ appearing in the tensor product has weight zero; hence it does not contribute to the weight of the basis elements. Also the $\mathrm{SL}_6(\mathbb{C})$ factor of our group acts trivially on F^*, and hence to obtain the weight of $f_1^* \otimes 1$ we concatenate `{0,0,0,0,0}`, the weight of the trivial representation of $\mathrm{SL}_6(\mathbb{C})$, with `{-1,0}`, the weight of f_1^*. We proceed similarly for the other basis vectors and record the weights in the list `U0`:

```
i14 : U0={{0,0,0,0,0,-1,0},{0,0,0,0,0,1,-1},{0,0,0,0,0,0,1}};
```

At this point we are ready to decompose BR. To do so, we issue the command `highestWeightsDecomposition` with three arguments: the first is BR, the second is an integer i informing Macaulay2 that we wish to provide the weights in the i-th free module of the complex, and the third is the list of weights in the coordinate basis of the i-th module (remember the indexing of the modules starts from zero in Macaulay2):

```
i15 : H0=highestWeightsDecomposition(BR,0,U0)

o15 = HashTable{0 => HashTable{{0} => Tally{{0, 0, 0, 0, 0, 0, 1} => 1}}}
                1 => HashTable{{1} => Tally{{1, 0, 0, 0, 0, 0, 0} => 1}}
                2 => HashTable{{4} => Tally{{0, 0, 0, 1, 0, 0, 0} => 1}}
                3 => HashTable{{5} => Tally{{0, 0, 0, 0, 1, 1, 0} => 1}}
                4 => HashTable{{6} => Tally{{0, 0, 0, 0, 0, 2, 0} => 1}}

o15 : HashTable
```

We deduce that BR decomposes as

$$F^* \otimes R \leftarrow E \otimes R(-1) \leftarrow \textstyle\bigwedge^4 E \otimes R(-4) \leftarrow \textstyle\bigwedge^5 E \otimes F \otimes R(-5) \leftarrow \mathbb{S}_2 F \otimes R(-6) \leftarrow 0.$$

If we choose to start from the second module, we need to provide the list of weights of the elements $e_1 \otimes 1, \dots, e_6 \otimes 1$. The commands are:

```
i16 : U1={{1,0,0,0,0,0,0},{-1,1,0,0,0,0,0},{0,-1,1,0,0,0,0},{0,0,-1,1,0,0,0},
        {0,0,0,-1,1,0,0},{0,0,0,0,-1,0,0}};

i17 : H1=highestWeightsDecomposition(BR,1,U1); H0===H1

o18 = true
```

Indeed, the decomposition is the same.

As with rings and ideals, we can decompose the graded components of a module. The difference is that we need to provide a list of weights for the generators of the presentation used to define the module. For our module M, this is exactly the list `U0` introduced earlier. As usual, we may decompose a single degree or a range.

```
i19 : highestWeightsDecomposition(M,-1,2,U0)

o19 = HashTable{-1 => Tally{}                            }
                 0 => Tally{{0, 0, 0, 0, 0, 0, 1} => 1}
                 1 => Tally{{1, 0, 0, 0, 0, 1, 1} => 1}
                 2 => Tally{{0, 1, 0, 0, 0, 0, 2} => 1}
                            {2, 0, 0, 0, 0, 2, 1} => 1

o19 : HashTable
```

Since M is generated in degree zero, we see that the output contains an empty decomposition in degree -1. Whereas we see, for example, that

$$M_2 = \bigwedge^2 E \otimes \mathbb{S}_{2,2} F \oplus \mathbb{S}_2 E \otimes \mathbb{S}_{3,1} F.$$

4. CLOSING REMARKS. Here we comment on a few points of (potential) interest.

- The method `highestWeightsDecomposition` provides the main functionality of this package. This method relies on the method `propagateWeights` and the function `decomposeWeightsList`, both of which are also exported. The method `propagateWeights` implements (with minor changes) an algorithm of [Galetto 2015]. The function `decomposeWeightsList` implements a modified version of Freudenthal's multiplicity formula using the algorithm discussed in [de Graaf 2000, Chapter 8.9] and [Moody and Patera 1982]. We do not anticipate the user employing these commands directly, but they are available for those who wish to experiment with them. More details are available in the package documentation.
- Multigradings are supported as well as single gradings, whenever they are compatible with the group action. An example is included in the documentation that involves multigradings.
- Decomposing graded components of rings and modules tends to work better in low degrees, as the dimension of graded components can grow fast.
- All the examples presented in this article and in the documentation of this package are over the field $\mathbb{C}$ of complex numbers for representation-theoretic reasons; however, all computations are performed in Macaulay2 over the field $\mathbb{Q}$ of rational numbers. The reader interested in an explanation of why the computational results obtained over $\mathbb{Q}$ can be interpreted over $\mathbb{C}$ may consult [Galetto 2015].
- For further concrete examples where the package *HighestWeights* may be useful, including links to Macaulay2 files, see [Galetto 2014]. The Macaulay2 package *PieriMaps* [Sam 2009] provides means to construct additional examples.

APPENDIX: SOME REPRESENTATION THEORY. We present here a brief review of the representation theory of semisimple Lie groups with the terminology used in this paper and by the package *HighestWeights*. For more details the reader can consult [Humphreys 1975; 1978; Fulton and Harris 1991]. The manual of the software package *LiE* [van Leeuwen et al. 1992] also contains a brief, and in our opinion well-written, review of the theory.

All Lie groups are intended to be complex Lie groups.

A1. ***Type and representations of a semisimple Lie group.*** A (simply connected) simple Lie group is one of the following: the special linear group $\mathrm{SL}_{n+1}(\mathbb{C})$ (type A_n), the spin group $\mathrm{Spin}_{2n+1}(\mathbb{C})$ (type B_n) or $\mathrm{Spin}_{2n}(\mathbb{C})$ (type D_n), the symplectic group $\mathrm{Sp}_{2n}(\mathbb{C})$ (type C_n), or one of the exceptional groups of type E_6, E_7, E_8, F_4 and G_2. Every semisimple Lie group G is the quotient of a finite product of the simple Lie groups above, called the simple components of G, by a finite subgroup. The type of G is then obtained by concatenating the types of the simple components.

Let $\mathbb{C}^\times$ be the multiplicative group of nonzero complex numbers. A *torus* is a Lie group which is isomorphic to $(\mathbb{C}^\times)^n$, for some positive integer n called the rank of the torus. Every semisimple Lie group G contains a maximal torus. All maximal tori are conjugate and hence have the same rank; the rank of G is defined to be the rank of a maximal torus. Usually one particular maximal torus is fixed, and it is denoted by T. The character group of T, denoted by $X(T)$, is the set of all Lie group homomorphisms $\chi : T \to \mathbb{C}^\times$. For any finite-dimensional representation V of G, there is a unique decomposition $V = \bigoplus_{\chi \in X(T)} V_\chi$, where $V_\chi = \{v \in V \mid \forall t \in T, t \cdot v = \chi(t)v\}$. The characters χ such that $V_\chi \neq 0$ are called *weights* of V, and $\dim V_\chi$ is called the multiplicity of χ in V. Each subspace V_χ is called a weight space, and its nonzero elements are called weight vectors with weight χ. The weights of V along with their multiplicities uniquely determine V as a representation of G. Moreover, if T has rank n, there is a group isomorphism $X(T) \cong \mathbb{Z}^n$, and hence weights can be simply recorded as lists of integers.

Every semisimple Lie group G contains a maximal connected solvable subgroup B, called Borel subgroup, which contains the fixed maximal torus T. Let V be a finite-dimensional representation of G and let $v \in V$ be a weight vector of weight ω. If v spans a B-stable one-dimensional subspace of V, then v is said to be a *highest-weight vector*. The representation V is irreducible if and only if v is, up to multiplication by a scalar, the only highest-weight vector in V. In this case, V is the unique irreducible representation of G with highest weight ω, and it is often denoted by $V(\omega)$.

A simple Lie group S of rank n has n fundamental representations. The construction of fundamental representations for each type is detailed in [Fulton and

Harris 1991], and a summary can be found in [Tits 1967]. The concept of fundamental representations can be extended to any semisimple Lie group G: if G has simple components $S_1, \dots, S_r$, then for each component S_i and for each fundamental representation $V_{i,j}$ of S_i, take $V_{i,j}$ to be a fundamental representation of G with a trivial action of all the other components. Fundamental representations are irreducible, and their highest weights are called *fundamental weights*. The fundamental weights are often denoted $\omega_1, \dots, \omega_n$; they form a basis of $X(T)$ as a free abelian group, and the isomorphism $X(T) \cong \mathbb{Z}^n$ is usually taken to send them to the coordinate basis of $\mathbb{Z}^n$. From now on we will always write weights as elements of the additive group $\mathbb{Z}^n$, and always in the basis of fundamental weights.

Since the fundamental weights $\omega_1, \dots, \omega_n$ form a basis of $X(T)$, every weight $\omega \in X(T)$ can be written as $\omega = \sum_{i=1}^n m_i\omega_i$ for some $m_1, \dots, m_n \in \mathbb{Z}$. Computationally the weight ω is represented by the list of integers $(m_1, \dots, m_n)$. A weight $\omega = \sum_{i=1}^n m_i\omega_i$ is called *dominant* if $m_i \geqslant 0$ for all $i \in \{1, \dots, n\}$. There is a bijection between dominant weights of $X(T)$ and irreducible representations of G, which sends ω to the highest-weight representation $V(\omega)$. The identity element of $X(T)$, i.e., the zero element of $\mathbb{Z}^n$ represented by the list $(0, \dots, 0)$, is a dominant weight, and it corresponds to the trivial representation of G.

A2. ***The special linear group.*** The special linear group $\mathrm{SL}_{n+1}(\mathbb{C})$ is the set of complex matrices with determinant 1; it is a simple Lie group of type A_n. The subset of diagonal matrices in $\mathrm{SL}_{n+1}(\mathbb{C})$ forms a subgroup which is a maximal torus of rank n. The corresponding Borel subgroup is the subset of upper-triangular matrices in $\mathrm{SL}_{n+1}(\mathbb{C})$.

The fundamental representations of $\mathrm{SL}_{n+1}(\mathbb{C})$ are (in order) the exterior powers $\mathbb{C}^{n+1}, \bigwedge^2\mathbb{C}^{n+1}, \dots, \bigwedge^n\mathbb{C}^{n+1}$. Their weights are the fundamental weights $\omega_1, \dots, \omega_n$, and ω_i is represented by a list of zeroes with a 1 in the i-th entry.

The irreducible representations of $\mathrm{SL}_{n+1}(\mathbb{C})$ are given by the Schur modules $\mathbb{S}_\lambda\mathbb{C}^{n+1}$, where λ is a partition with at most n parts. For the construction of Schur modules the reader may consult [Fulton 1997, Chapter 8] or [Fulton and Harris 1991, Chapter 6]. The representation $\mathbb{S}_\lambda\mathbb{C}^{n+1}$ is the highest-weight representation $V(\omega)$ for the weight $\omega = (\lambda_1 - \lambda_2)\omega_1 + \cdots + (\lambda_{n-1} - \lambda_n)\omega_{n-1} + \lambda_n\omega_n$.

To fix an example, consider the group $\mathrm{SL}_4(\mathbb{C})$ of type A_3. The type of our group may be input by typing `DynkinType{{"A",3}}`. The representation $\mathbb{C}^4$, also known as standard representation, has highest weight `{1,0,0}`, being the same as the Schur module $\mathbb{S}_1\mathbb{C}^4$. The coordinate basis $\{e_1, e_2, e_3, e_4\}$ of $\mathbb{C}^4$ is a basis of weight vectors, and the weights of e_1, e_2, e_3, e_4 are `{1,0,0}`, `{-1,1,0}`, `{0,-1,1}` and `{0,0,-1}` respectively. If we tensor $\mathbb{C}^4$ with itself, we obtain a new representation of $\mathrm{SL}_4(\mathbb{C})$ with basis $e_i \otimes e_j$ for $i, j \in \{1, 2, 3, 4\}$. Moreover, the vector $e_i \otimes e_j$ is a weight vector with weight equal to the weight of e_i plus the weight of e_j. The same principle of adding weights applies to symmetric and antisymmetric tensors, as well as to the tensor product of two different representations of the same group. For example, the weight of $e_1 \wedge e_2$ in $\bigwedge^2 \mathbb{C}^4$ is `{0,1,0}`. Indeed, $\bigwedge^2 \mathbb{C}^4$ is a highest-weight representation with highest weight `{0,1,0}`, thus corresponding to the Schur module $\mathbb{S}_{1,1}\mathbb{C}^4$, and $e_1 \wedge e_2$ is its highest-weight vector. Let us also mention the dual representation $(\mathbb{C}^4)^*$; the elements of the dual basis $e_1^*, e_2^*, e_3^*, e_4^*$ are weight vectors with weights `{-1,0,0}`, `{1,-1,0}`, `{0,1,-1}` and `{0,0,1}`. Note that the weight of e_i^* is the additive inverse of the weight of e_i. Here the only highest weight is `{0,0,1}`, so $(\mathbb{C}^4)^*$ is isomorphic to $\bigwedge^3 \mathbb{C}^4$.

When dealing with a (quotient of a) product of groups, say $\mathrm{SL}_4(\mathbb{C}) \times \mathrm{SL}_6(\mathbb{C})$, we declare the type, listing in order the types of the simple components like this: `DynkinType{{"A",3},{"A",5}}`. The representations of $\mathrm{SL}_4(\mathbb{C}) \times \mathrm{SL}_6(\mathbb{C})$ are obtained by tensoring a representation of $\mathrm{SL}_4(\mathbb{C})$ with one of $\mathrm{SL}_6(\mathbb{C})$. Their lists of weights are then concatenated to form a single list. For example, the representation $\mathbb{S}_{6,3,1}\mathbb{C}^4 \otimes \bigwedge^2 \mathbb{C}^6$ has highest weight `{3,2,1,0,1,0,0,0}`, because the highest weights of $\mathbb{S}_{6,3,1}\mathbb{C}^4$ and $\bigwedge^2 \mathbb{C}^6$ are `{3,2,1}` and `{0,1,0,0,0}` respectively. Similarly, the irreducible representation with highest weight `{0,0,0,1,1,1,1,0}` is the tensor product of the irreducible representations $\mathbb{C}$ of $\mathrm{SL}_4(\mathbb{C})$ and $\mathbb{S}_{4,3,2,1}\mathbb{C}^6$ of $\mathrm{SL}_4(\mathbb{C})$, which is simply isomorphic to $\mathbb{S}_{4,3,2,1}\mathbb{C}^6$.

ACKNOWLEDGEMENTS. The author wishes to thank Jianyun Shan for his help running the first example (mostly by hand!), Luke Oeding for submitting some interesting test examples and Adam Boocher for some helpful suggestions on the documentation. The author is indebted to the entire Macaulay2 community for its support; a special thanks goes to Mike Stillman and Dan Grayson, and to Baptiste Calmès and Viktor Petrov, whose Macaulay2 package *WeylGroups* facilitates many of the background computations carried out by our package. Finally, the author thanks the anonymous referees for providing many useful comments and helping to improve the quality of this article, as well as the source code and documentation of this package.

SUPPLEMENT. The online supplement contains version 0.6.5 of *HighestWeights*.

REFERENCES.

[Eisenbud 1995] D. Eisenbud, *Commutative algebra*, Graduate Texts in Mathematics **150**, Springer, New York, 1995. MR 97a:13001 Zbl 0819.13001

[Fulton 1997] W. Fulton, *Young tableaux*, London Mathematical Society Student Texts **35**, Cambridge University Press, 1997. MR 99f:05119 Zbl 0878.14034

[Fulton and Harris 1991] W. Fulton and J. Harris, *Representation theory*, Graduate Texts in Mathematics **129**, Springer, New York, 1991. MR 93a:20069 Zbl 0744.22001

[Galetto 2014] F. Galetto, "Free resolutions of orbit closures for the representations associated to gradings on Lie algebras of type E_6, F_4 and G_2", preprint, 2014. arXiv 1210.6410

[Galetto 2015] F. Galetto, "Propagating weights of tori along free resolutions", *J. Symb. Comput.* (2015). doi 10.1016/j.jsc.2015.05.004.

[de Graaf 2000] W. A. de Graaf, *Lie algebras: theory and algorithms*, North-Holland Mathematical Library **56**, North-Holland, Amsterdam, 2000. MR 2001j:17011 Zbl 1122.17300

[Grayson and Stillman] D. R. Grayson and M. E. Stillman, "Macaulay2: a software system for research in algebraic geometry", available at http://www.math.uiuc.edu/Macaulay2.

[Humphreys 1975] J. E. Humphreys, *Linear algebraic groups*, Graduate Texts in Mathematics **21**, Springer, New York-Heidelberg, 1975. MR 53 #633 Zbl 0325.20039

[Humphreys 1978] J. E. Humphreys, *Introduction to Lie algebras and representation theory*, Graduate Texts in Mathematics **9**, Springer, New York-Berlin, 1978. Revised 2nd printing. MR 81b:17007 Zbl 0447.17001

[Kreuzer and Robbiano 2005] M. Kreuzer and L. Robbiano, *Computational commutative algebra, II*, Springer, Berlin, 2005. MR 2006h:13036 Zbl 1090.13021

[van Leeuwen et al. 1992] M. van Leeuwen, A. Cohen, and B. Lisser, "LiE, A Package for Lie Group Computations", 1992, available at http://www-math.univ-poitiers.fr/~maavl/LiE/.

[Moody and Patera 1982] R. V. Moody and J. Patera, "Fast recursion formula for weight multiplicities", *Bull. Amer. Math. Soc.* (*N.S.*) **7**:1 (1982), 237–242. MR 84a:17005 Zbl 0494.17005

[Sam 2009] S. V. Sam, "Computing inclusions of Schur modules", *J. Softw. Algebra Geom.* **1** (2009), 5–10. MR 2878669 Zbl 1311.13039

[Shafarevich 1994] I. R. Shafarevich, *Basic algebraic geometry, I: Varieties in projective space*, 2nd ed., Springer, Berlin, 1994. MR 95m:14001 Zbl 0797.14001

[Tits 1967] J. Tits, *Tabellen zu den einfachen Lie Gruppen und ihren Darstellungen*, Springer, Berlin-New York, 1967. MR 36 #1575 Zbl 0166.29703

RECEIVED: 11 Jul 2014 REVISED: 23 Aug 2014 ACCEPTED: 5 Jun 2015

FEDERICO GALETTO:

galetto.federico@gmail.com
Department of Mathematics and Statistics, Queen's University, 48 University Avenue, Kingston K7L 3N6, Canada

Journal of Software for Algebra and Geometry vol 7, no 1, 2015